Key Homework 15

Math 277, Fall 2005

Ordinary Differential Equations

Dr. Goutziers

Initializations

```
> restart;
  with(linalg):
```

Warning, the protected names norm and trace have been redefined and unprotected

19: Page 167

Code the differential equation and compute the auxiliary equation.

```
> deq:=diff(y(t), t$2)-4*diff(y(t), t)-5*y(t)=0;
```

$$deq := \left(\frac{d^2}{dt^2}\,\mathrm{y}(t)\right) - 4\left(\frac{d}{dt}\,\mathrm{y}(t)\right) - 5\,\mathrm{y}(t) = 0$$

```
> aux_eq:=simplify(eval(subs(y(t)=exp(r*t), deq))/exp(r*t));
```

$$aux\_eq := r^2 - 4\,r - 5 = 0$$

Solve the auxiliary equation to find the eigenvalues.

```
> evals:=solve(aux_eq, r);
```

$$evals := 5, -1$$

Since the eigenvalues are real and distinct, the corresponding eigenfunctions are given by.

```
> for k to 2 do
  Y[k]:=exp(evals[k]*t);
  od;
```

$$Y_1 := \mathbf{e}^{(5\,t)}$$

$$Y_2 := \mathbf{e}^{(-t)}$$

Use the Wronskian to check for linear independence.

```
> check:=simplify(det(wronskian([Y[1],Y[2]], t)));
```

$$check := -6\,\mathbf{e}^{(4\,t)}$$

Since the Wronskian is never zero, we can be sure that Y_1 and Y_2 are linearly independent. Therefore, the general solution is given by

```
> gensol:=add(c[k]*Y[k], k=1..2);
```

$$gensol := c_1\,\mathbf{e}^{(5\,t)} + c_2\,\mathbf{e}^{(-t)}$$

Finally, implement the initial conditions and solve the initial value problem.

```
> eq1:=subs(t=-1, gensol)=3;
  eq2:=subs(t=-1, diff(gensol, t))=9;
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1\,\mathbf{e}^{(-5)} + c_2\,\mathbf{e} = 3$$

$$eq2 := 5\,c_1\,\mathbf{e}^{(-5)} - c_2\,\mathbf{e} = 9$$

$$val\_c := \{c_2 = \frac{1}{\mathbf{e}},\, c_1 = \frac{2}{\mathbf{e}^{(-5)}}\}$$

```
> ans[19]:=simplify(subs(val_c, gensol));
```

$$ans_{19} := 2\,\mathbf{e}^{(5+5\,t)} + \mathbf{e}^{(-1-t)}$$

```
>
```

20: Page 167

Code the differential equation and compute the auxiliary equation.

```
> deq:=diff(y(t), t$2)-4*diff(y(t), t)+4*y(t)=0;
```

$$deq := \left(\frac{d^2}{dt^2} y(t)\right) - 4\left(\frac{d}{dt} y(t)\right) + 4\, y(t) = 0$$

```
> aux_eq:=simplify(eval(subs(y(t)=exp(r*t), deq))/exp(r*t));
```

$$aux\_eq := r^2 - 4\, r + 4 = 0$$

Solve the auxiliary equation to find the eigenvalues.

```
> evals:=solve(aux_eq, r);
```

$$evals := 2, 2$$

Since 2 is a double root of the auxiliary equation, the corresponding eigenfunctions are given by.

```
> Y[1]:=exp(evals[1]*t);
  Y[2]:=t*Y[1];
```

$$Y_1 := \mathbf{e}^{(2\,t)}$$

$$Y_2 := t\, \mathbf{e}^{(2\,t)}$$

Use the Wronskian to check for linear independence.

```
> check:=simplify(det(wronskian([Y[1],Y[2]], t)));
```

$$check := \mathbf{e}^{(4\,t)}$$

Since the Wronskian is never zero, we can be sure that Y_1 and Y_2 are linearly independent. Therefore, the general solution is given by

```
> gensol:=add(c[k]*Y[k], k=1..2);
```

$$gensol := c_1\, \mathbf{e}^{(2\,t)} + c_2\, t\, \mathbf{e}^{(2\,t)}$$

Finally, implement the initial conditions and solve the initial value problem.

```
> eq1:=subs(t=1, gensol)=1;
  eq2:=subs(t=1, diff(gensol, t))=1;
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1\, \mathbf{e}^2 + c_2\, \mathbf{e}^2 = 1$$

$$eq2 := 2\, c_1\, \mathbf{e}^2 + 3\, c_2\, \mathbf{e}^2 = 1$$

$$val\_c := \{c_2 = -\frac{1}{\mathbf{e}^2},\, c_1 = \frac{2}{\mathbf{e}^2}\}$$

```
> ans[20]:=simplify(subs(val_c, gensol));
```

$$ans_{20} := -\mathbf{e}^{(-2+2\,t)}\, (-2+t)$$

```
>
```

44: Page 169

Code the differential equation and generate the auxiliary equation.

```
> deq:=diff(y(t), t$3)-7*diff(y(t), t$2)+7*diff(y(t), t)+15*y(t)=0;
```

$$deq := \left(\frac{d^3}{dt^3} y(t)\right) - 7\left(\frac{d^2}{dt^2} y(t)\right) + 7\left(\frac{d}{dt} y(t)\right) + 15\, y(t) = 0$$

```
> aux_eq:=simplify(eval(subs(y(t)=exp(r*t), deq))/exp(r*t));
```

$$aux\_eq := r^3 - 7\, r^2 + 7\, r + 15 = 0$$

Solve the auxiliary equation and find the eigenvalues.

```
> evals:=solve(aux_eq, r);
```

$$evals := -1, 3, 5$$

Code the corresponding solutions.

```
> for k to 3 do
  Y[k]:=exp(evals[k]*t);
  od;
```

$$Y_1 := \mathbf{e}^{(-t)}$$

$$Y_2 := e^{(3\,t)}$$

$$Y_3 := e^{(5\,t)}$$

Check for linear independence.

```
> check:=simplify(det(wronskian([seq(Y[k], k=1..3)], t)));
```

$$check := 48\, e^{(7\,t)}$$

Since the Wronskian is never zero, these solutions are linearly independent. Hence the general solution is given by

```
> ans[44]:=add(c[k]*Y[k], k=1..3);
```

$$ans_{44} := c_1\, e^{(-t)} + c_2\, e^{(3\,t)} + c_3\, e^{(5\,t)}$$

```
>
```

49 (a): Page 169

Code the differential equation and generate the auxiliary equation.

```
> deq:=3*diff(y(t), t$3)+18*diff(y(t), t$2)+13*diff(y(t), t)-19*y(t)=0;
```

$$deq := 3\left(\frac{d^3}{dt^3}\,y(t)\right) + 18\left(\frac{d^2}{dt^2}\,y(t)\right) + 13\left(\frac{d}{dt}\,y(t)\right) - 19\,y(t) = 0$$

```
> aux_eq:=simplify(eval(subs(y(t)=exp(r*t), deq))/exp(r*t));
```

$$aux\_eq := 3\,r^3 + 18\,r^2 + 13\,r - 19 = 0$$

Solve the auxiliary equation numerically and find the eigenvalues.

```
> evals:=fsolve(aux_eq, r);
```

$$evals := -4.831922263, -1.869273823, 0.7011960861$$

Code the corresponding solutions.

```
> for k to 3 do
  Y[k]:=exp(evals[k]*t);
  od;
```

$$Y_1 := e^{(-4.831922263\,t)}$$

$$Y_2 := e^{(-1.869273823\,t)}$$

$$Y_3 := e^{(0.7011960861\,t)}$$

Check for linear independence.

```
> check:=simplify(det(wronskian([seq(Y[k], k=1..3)], t)));
```

$$check := 42.13690210\, e^{(-6.\,t)}$$

Since the Wronskian is never zero, these solutions are linearly independent. Hence the general solution is given by

```
> ans[`49(a)`]:=add(c[k]*Y[k], k=1..3);
```

$$ans_{49(a)} := c_1\, e^{(-4.831922263\,t)} + c_2\, e^{(-1.869273823\,t)} + c_3\, e^{(0.7011960861\,t)}$$

```
>
```